The Program: (Handout)
  /*********************************************************/

  /* DYNAMIC ALLOCATION OF NEW DATASETS USING DATASET NAMES*/

  /* CONSTRUCTED ON THE FLY DEPENDING ON HOW MANY OUTPUTS  */

  /* WILL BE REQUIRED BY THE DATA. THIS PROGRAM WILL CREATE*/

  /* SAS CODE ON THE FLY, WRITE IT TO SASPARM, THEN        */

  /* %INCLUDE SASPARM TO EXECUTE THE CODE THAT WAS JUST    */

  /* CREATED.                                              */

  /*                                                       */

  /* WARNING: USE THIS FUNCTIONALITY WITH CAUTION!  IT IS  */

  /* POSSIBLE TO ALLOCATE TOO MANY DATASETS IN A SINGLE    */

  /* JOBSTEP AND ABEND THE WHOLE THING. KEEP THE NUMBER OF */

  /* NEW DATASETS CREATED TO FEWER THAN 50 TO BE SAFE.     */

  /*********************************************************/

   OPTIONS SOURCE SOURCE2 SORTDEV=3390;

  /**************************************/

  /* THIS DATASTEP READS THE DATA FILE. */

   DATA REP_DATA;

        INFILE INDD1;

        INPUT @ 01  REPID  $3.

              @ 05  NAME   $6.

              @ 14  STUFF  $CHAR16.;

   RUN; 

  /**************************************/

  /* CREATE LIST OF UNIQUE REPID VALUES.*/

   PROC FREQ DATA=REP_DATA;

        TABLES REPID/LIST NOPRINT NOPERCENT OUT=REPID_LIST;

   RUN;

  /* (continued below) */

  /*****************************************************/

  /* THIS DATASTEP WILL CREATE A LIST OF DATASET NAMES */

  /* AND FILEREFS TO USE WHEN WRITING OUT TO THE FILES */

   DATA DSN_LIST;

        LENGTH FILE_REF $ 8;

        LENGTH OSDSN $ 45;

        LENGTH MSDSN $ 45;

       /* CREATE A DATASET NAME AND FILEREF FOR EACH UNIQUE REPID */

        SET REPID_LIST;

        FILE_REF = 'FIL' || REPID;

       /* MAINFRAME FORMAT */

        OSDSN = '&sysuid.OMSC.DAILY.REPDATA.REP' || REPID;

       /* WINDOWS FORMAT */

        MSDSN = 'C:\My Documents\Jim Noe\OMSC\RepData\Rep' || REPID;

       /* IF THE NUMBER OF OUTPUTS IS LIMITED THEN THIS STATEMENT */

       /* WILL STOP EXECUTING THIS DATASTEP AT THE LIMIT NUMBER.  */

        IF _N_ > 49

        THEN STOP;

   RUN;

  /*********************************************************/

  /* PRINT DATASET NAME LIST TO VERIFY NAMES CONSTRUCTED OK*/

   PROC PRINT DATA=DSN_LIST;

   RU;

  /* (continued below) */ 

  /********************************************************************/

  /* THIS DATASTEP WILL OUTPUT THE FILENAME STATEMENTS IN THE FORMAT  */

  /*  "FILENAME FIL### '&sysuid.TEST.FILE###' DISP=(NEW,CATLG)"       */

  /* PLUS THE LRECL= AND BLKSIZE= PARAMETERS TO THE SASPARM OUTPUT.   */

   DATA _NULL_;

        LENGTH LINEOUT $ 72;

        SET DSN_LIST END=_EOF;

        FILE SASPARM NOTITLES NOPRINT;  /* TO WRITE TO OUTPUT SASPARM */

     /* WRITE MAINFRAME FORMAT OF FILENAME STATEMENT */

        LINEOUT = "FILENAME " || TRIM(FILE_REF) || " '"

                              || TRIM(OSDSN)    ||

                  "' DISP=(NEW,CATLG) ";

        PUT @ 01  LINEOUT;

        PUT @ 10  "RECFM=FB LRECL=100 BLKSIZE=27900;";

     /* WRITE WINDOWS FORMAT OF FILENAME STATEMENT */

     /*                                                    */

     /* LINEOUT = "FILENAME " || TRIM(FILE_REF) || " '"    */

     /*                       || TRIM(MSDSN)    || "';"    */

     /* PUT @ 01  LINEOUT;                                 */

     /* ADD WHATEVER ELSE YOU NEED ON THE WINDOWS PLATFORM.*/

        PUT @ 05  " ";                  /* WRITE A BLANK LINE */

        IF _EOF

        THEN PUT @ 01 "RUN;";

   RUN;

  /************************************************************/

  /* THIS %INCLUDE WILL COPY IN THE STATEMENTS AND EXECUTE    */

  /* THEM. THE FILEREFS ARE DEFINED, DATASETS ALLOCATED AND   */

  /* THEY ARE READY TO HAVE DATA WRITTEN TO THEM.             */

   %INCLUDE SASPARM;

  /* (continued below) */ 

  /************************************************************/

  /* THE DATASTEP THAT FOLLOWS WILL CREATE A SAS IF STATEMENT */

  /* THAT ASSIGNS THE OUTPUT FILE TO WHICH DATA IS TO BE      */

  /* WRITTEN DEPENDING ON THE VARIABLE NAMED REPID.           */

   DATA _NULL_;

        LENGTH LINEOUT $ 72;

        SET DSN_LIST;  /* CREATED AT TOP OF THIS PROGRAM */

        FILE SASPARM NOTITLES NOPRINT;  /* TO WRITE TO OUTPUT SASPARM */

      /* CREATE 'IF' STATEMENT THAT DECIDES WHERE TO WRITE EACH */

      /* OBSERVATION DEPENDING ON THE VARIABLE SELECT_FILE.     */

        IF _N_ = 1

        THEN PUT @ 05 "IF REPID = " REPID;

        ELSE PUT @ 05 "ELSE IF REPID = " REPID;

      /* DEFINE EACH OUTPUT FILEREF */

        LINEOUT = "THEN FILE "

                  || TRIM(FILE_REF) || " NOTITLES NOPRINT;";

        PUT @ 05  LINEOUT;

   RUN;

  /***********************************************************/

  /* THIS DATASTEP READS YOUR DATA FILE, DECIDES WHICH REPID */

  /* IS WRITTEN TO WHICH OUTPUT FILE, ASSIGNS THE OUTPUT     */

  /* FILEREF AND THEN WRITES OUT ONE OBSERVATION'S WORTH OF  */

  /* DATA TO THE OUTPUT FILE SELECTED.                       */

   DATA _NULL_;

        LENGTH LINEOUT $ 72;

        SET REP_DATA;

  /* THIS INCLUDE COPIES IN THE 'IF' STATEMENT THAT ASSIGNS   */

  /* AN OUTPUT FILEREF DEPENDING ON THE VALUE OF SELECT_FILE. */

        %INCLUDE SASPARM;

  /* THESE PUT STATEMENTS WRITE OUT THE CONTENTS OF THE FILE.  */

        LINEOUT = "DEAR " || TRIM(NAME) || ",";

        PUT @ 01  LINEOUT;

        PUT @ 01  ' ';

        LINEOUT = "YOUR REPID IS " || TRIM(REPID) || ".";

        PUT @ 01  LINEOUT;

        PUT @ 01  STUFF;

   RUN;

  /*************************************************************/

  /***************** END OF SAMPLE PROGRAM *********************/

  /*************************************************************/

  /* Note: The statements created dynamically are not included */

  /* here because they ARE dynamic and depend on input values. */



























































































